

NASCOM 280 NYT

NASCOM BRUGERGRUPPE
2730 Herlev

Sidevolden 23
Giro 6742602

NR: 8
2. årgang

September 1981.

Sammen med dette nummer udkommer medlemsliste og programbiblioteksliste, derfor denne "tynde" udgave af NN.

Jeg skal pointere, at medlemslisten er udfærdiget den 1.9.81 og den derfor ikke indeholder navnene på de der først har betalt kontingent efter den 1.9. Enhver ændring som du gerne vil have foretaget i listen bedes du indsende til min adresse, da jeg fører medlemslisten, mens Erik Hansen fører adresselisten, hvorefter NN udsendes.

Det kan ikke kraftigt nok udtrykkes, at bladet også er dit blad og ikke alene redaktørens. Jeg opfordre til, at du indsender alt hvad du kunne tænke dig kom med i bladet. Jeg har således for at få en bredere dækning bedt samtlige bestyrelsesmedlemmer om et bidrag til næste nummer, hvor jeg vil holde mig diskret tilbage. Hvis der er enkelte, der vil indgå i en redaktionsgruppe eller der er nogen, der vil hjælpe med indtastningen af tekst i Naspen-format, bedes de kontakte mig snarest.

I håb om mange gode timer ønsker jeg et god sæson

Sir Asbjørn.

Indhold

=====

side 2	Matrixregning fortsat (Basic)
side 4	Intelligente maskiner (Basic)
side 6	Salg

Matrixregning.

Vi så sidste gang, hvordan man kunne addere, subtrahere og multiplicere. Her vil jeg da forsøge at vise, hvad det kan bruges til.

Er det almindeligt også uden for matematiske problemer at bruge matrixer? Hertil må man svare, at styrken netop ligger i analytisk geometri, men hver gang man sammenkobler i en tabel et sæt oplysninger, gør man brug af en matrix. Det kan f.eks. være en længdetabel mellem danske købstæder, eller en tabel over lagerindhold, antal og pris.

Her vil jeg uddybe en gren af geometrien - løsning af ligningssystemer. Hver gang man kan opstille en sammenhæng mellem nogle forskellige variable, kan man benytte matrixregning som løsningsmiddel. Man skal sørge for, at der er lige mange ubekendte som ligninger (hvem kan ikke huske skolelærdommen: 2 ligninger med to ubekendte - eller - 3 ligninger med 3 ubekendte).

Det kan hende, der ikke er nogen løsning til opgaven, fordi den geometriske tolkning viser, at "linierne" er parallelle. Dette tilfælde kan undersøges v.h.a. en determinantundersøgelse, men her henvises til litteraturen.

Et ligningssystem kan opstilles helt generelt som følger:

$$\begin{array}{r} a_{11}x_1 + a_{12}x_2 + \dots + a_{1i}x_i = b_1 \\ a_{21}x_1 + a_{22}x_2 + \dots + a_{2i}x_i = b_2 \\ \vdots \\ a_{i1}x_1 + a_{i2}x_2 + \dots + a_{ii}x_i = b_i \end{array}$$

x_1, x_2, \dots, x_i er de ubekendte og vi kan skrive dem på følgende måde: $A \cdot X = B$, hvor A er koefficienterne til de ubekendte. X er vektoren af ubekendte og B er b_1, b_2, \dots, b_i .

Denne ligning har løsningen $X = (A^{-1}) \cdot B$, dvs at A inverteres og derefter multipliceres med B . Dette betyder, at vi bliver nødt til at "dyrke" en matrix inverterede form. Definitionen er at $(A^{-1}) \cdot A = E$, hvor E er en enhedsmatrix, i hvilken alle elementerne er 0, undtagen den ene skrålinje, som er 1. F.eks. som følger (matrixen skal være kvadratisk for at kunne inverteres):

$$\begin{bmatrix} 1 & 0 & 0 & 0 & 0 \\ 0 & 1 & 0 & 0 & 0 \\ 0 & 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 & 0 \\ 0 & 0 & 0 & 0 & 1 \end{bmatrix} = E$$

Følgende program inverterer en vilkårlig, kvadratisk matrix:

```

100 PRINT"LØSNING AF LIGNINGSSYSTEMER MED, HVOR"
110 INPUT"MANGE UBEKENDTE";R
120 PRINT"MATRIX A"
130 FOR T1=1 TO R
140 PRINT"SØJLE";T1.
150 FOR T2=1 TO R
160 INPUT A(T2,T1)
170 NEXT T2
180 B(T1,T1)=1
190 NEXT T1
200 PRINT
210 PRINT"INDTAST VEKTOR B"
220 FOR T=1 TO R
230 INPUT C(T)
240 NEXT T
250 FOR T1=1 TO R
260 FOR T2=T1 TO R
270 IF A(T2,T1)<>0 THEN GOTO310
280 NEXT T2
290 PRINT"DET LIGNINGSSYSTEM KAN IKKE LØSES !"
300 END
310 FOR T3=1 TO R
320 DU=A(T1,T3):A(T1,T3)=A(T2,T3):A(T2,T3)=DU
330 DU=B(T1,T3):B(T1,T3)=B(T2,T3):B(T2,T3)=DU
340 NEXT T3
350 IN=1/A(T1,T1)
360 FOR T3=1 TO R
370 A(T1,T3)=IN*A(T1,T3)
380 B(T1,T3)=IN*B(T1,T3)
390 NEXT T3
400 FOR T4=1 TO R
410 IF T4=T1 THEN470
420 IN=-A(T4,T1)
430 FOR T3=1 TO R
440 A(T4,T3)=A(T4,T3)+IN*A(T1,T3)
450 B(T4,T3)=B(T4,T3)+IN*B(T1,T3)
460 NEXT T3
470 NEXT T4
480 NEXT T1
490 PRINT
500 PRINT"LIGNINGSSYSTEMET HAR FØLGENDE LØSNING:"
510 FOR T1=1 TO R
520 AP=0
530 FOR T2=1 TO R
540 AP=AP+B(T1,T2)*C(T2)
550 NEXT T2
560 PRINT "X" T1=" AP
570 NEXT T1

```

Det næste program kan løse et vilkårligt ligningssystem, men husk at indtast et 0, hvis der ikke findes en koefficient til xi

```

100 INPUT "MATRIXINVERTERING AF STØRRELSE";R
110 INPUT"PLADSER EFTER KOMMAET (0-6)";A
120 AF=10^A
130 PRINT"MATRIX .A"
140 FOR T1=1 TO R
150 PRINT"SØJLE";T1
160 FOR T2=1 TO R
170 INPUT A(T2,T1)
180 NEXT T2
190 B(T1,T1)=1
200 NEXT T1
210 FOR T1=1 TO R
220 FOR T2=T1 TO R
230 IF A(T2,T1)<>0 THEN GOTO270
240 NEXT T2
250 PRINT"DENNE MATRIX KAN IKKE INVERTERES"
260 END
270 FOR T3=1 TO R
280 DU=A(T1,T3):A(T1,T3)=A(T2,T3):A(T2,T3)=DU
290 DU=B(T1,T3):B(T1,T3)=B(T2,T3):B(T2,T3)=DU
300 NEXT T3
310 IN=1/A(T1,T1)
320 FOR T3=1 TO R
330 A(T1,T3)=IN*A(T1,T3)
340 B(T1,T3)=IN*B(T1,T3)
350 NEXT T3
360 FOR T4=1 TO R
370 IF T4=T1 THEN430
380 IN=-A(T4,T1)
390 FOR T3=1 TO R
400 A(T4,T3)=A(T4,T3)+IN*A(T1,T3)
410 B(T4,T3)=B(T4,T3)+IN*B(T1,T3)
420 NEXT T3
430 NEXT T4
440 NEXT T1
450 PRINT
460 PRINT"INVERTERET MATRIX (A-1):"
470 FOR T1=1 TO R
480 FOR T2=1 TO R
490 PRINT INT(B(T1,T2)*AF+.5)/AF;
500 NEXT T2
510 PRINT
520 NEXT T1
Ok

```

Intelligente maskiner.

=====

Fra en lærebog i basic beregnet for studerende ved Roskilde universitetscenter har jeg sakset følgende program, der viser, hvordan man kan oplære en datamaskine. Udover afsnittet med datalogiske modeller, er der en indføring på ganske almindelig dansk med mange gode, velskrevne og veldokumenterede afsnit fra det helt simple til det mere avancerede. (ref. 1 - kan lånes på biblioteket).

Til demonstration bruges en simpel form for NIM, hvor der ligger 15 pinde på et bord og spillerne skiftes til at trække mellem 1 og 3 pinde. Det vil hurtigt vise sig, hvilke træk der er gode og hvilke, der er dårlige. Men den der starter vil efterhånden altid kunne vinde.

Programmet modspiller (i linie 2000) er programmet til at vælge det bedste træk eller vælge et tilfældigt, hvis det gode træk netop er "0" pinde, som man jo ikke må.

Begynder (hoved)programmet skal lære af sine erfaringer, så derfor benyttes en erfaringstabel E med 15 rækker og 3 søjler. Denne tabel skal man kunne slå op i for at se, hvor godt det påtænkte træk er. Til at begynde med står der "0" i hele tabellen, undtagen i de direkte ulovlige og tabsgivende træk.

antal pinde tilbage	antal pinde der trækkes		
	1	2	3
1	-1	-1	-1
2	0	-1	-1
3	0	0	-1
4	0	0	0
5	0	0	0
.	.	.	.
.	.	.	.
15	0	0	0

For at kunne holde rede på de gode træk under spillet, indføres de forskellige træk i den endimensionale tabel F med 15 elementer. Når spillet er slut spores resultatet i F. Hvis maskinen vandt, må positive elementer i F betragtes som gode, hvis det taber må vurderingen være omvendt. Denne vurdering af F overføres nu i E således: Alle elementer i E forøges med 1, hvor F angiver gode træk og mindskes med 1, hvor F angiver dårlige træk.

Herved opnås at visse træk (de gode) vokser i værdi, men de dårlige bliver mindre og mindre. Herved bliver programmet bedre og bedre til at spille NIM - for til sidst ikke er til at slå!!

ref.: Programmering og problemløsning
-grundbog i datalære

H. B. Hansen, Ole Caprani, Frank Jensen

/A.


```

10 REM PROG. LÆRER AT SPILLE EN SIMPEL FORM NIM
20 REM 15 PINDE LÆGGES. TO SPILLERE SKIFTES TIL
30 REM AT TAGE FRA BUNKEN - MINDST 1 OG HØJST
40 REM 3 PINDE. DEN DER TAGER SIDSTE PIND TABER
50 REM HOVEDPROGRAMMET SPILLER MOD ET UNDERPROG.
55 REM I LINIE 2000. HVOR DER OGSÅ KAN INDSET-
60 REM TES EN INPUTSÆTNING TIL MENNESKELIG MODS
70 REM E(15,3) EN ERFARINGSTABEL FOR ALLE TRÆK
80 REM F(15) FORLØBET AF DET AKTUELLE SPIL
90 REM N ANTAL SPIL I EN SPILSERIE
100 REM S SPILNR. I EN SPILSERIE
110 REM P ANTAL PINDE I BUNKEN
120 REM K PINDE DER FJERNES I TRÆKKET
130 REM T ANTAL TABTE SPIL I HOVEDPROG.
140 REM V INDIKATION AF VINDER
150 REM +1=HOVEDPROG., -1=UNDERPROG.
160 REM M DET HIDTIL BEDSTE TRÆKS GODHED
170 REM I,J HJÆLPEVARIABLE
180 REM H.B.HANSEN JANAUER 1977
190 REM
200 DIM E(15,3),F(15)
210 REM TILDELING AF NEUTRALE STARTVERDIER
220 FOR I=1 TO 15
230 F(I)=0
240 FOR J=1 TO 3
250 E(I,J)=0
260 NEXT J
270 NEXT I
280 REM MARKER UMULIGE TRÆK OG OPLAGTE TABSTREK
290 FOR I=1 TO 3
300 FOR J=1 TO 3
310 E(I,J)=-1
320 NEXT J
330 NEXT I
400 REM START NY SPILLE SERIE
410 PRINT"HVOR MANGE SPIL SKAL DER SPILLES";
420 INPUT N
430 IF N<=0 THEN END
440 REM START REGNSKAB MED TABTE SPIL
450 T=0
460 FOR S=1 TO N
470 REM START NYT SPIL
480 P=15
500 REM DET ER HOVEDPROGRAMMETS TUR. DET BEDSTE
510 REM TRÆK TIL P PINDE FINDES.
520 K=1
530 M=E(P,K)
540 FOR J=2 TO 3
550 IF M>=E(P,J) THEN 580
560 K=J
570 M=E(P,K)
580 NEXT J
590 REM MAN KAN IKKE TAGE FLERE PINDE END DER ER
600 IF P>=K THEN 620
610 K=P
620 REM HOVEDPROG. TAGER K PINDE
630 F(P)=K
640 P=P-K
650 REM DET UNDERSØGES OM SPILLET ER FORBI

```

```

660 IF P>0 THEN 710
670 REM SPILLET ER FORBI,UNDERPROGRAM VANDT
680 V=-1
690 T=T+1
700 GOTO 1000
710 REM DET ER UNDERPROGRAMMETS TUR
720 GOSUB 2000
730 REM MAN KAN IKKE TAGE FLERE PINDE END DER ER
740 IF P>=K THEN 760
750 K=P
760 REM UNDERPROGRAMMET TAGER K PINDE
770 F(P)=-K
780 P=P-K
790 REM DET UNDERSØGES OM SPILLET ER FORBI
800 IF P>0 THEN 500
810 REM SPILLET ER FORBI, OG HOVEDPROGRAM VANDT
820 V=1
1000 REM ERFARINGERNE UDDRAGES AF SPILLS FORLØB
1010 FOR I=1 TO 15
1020 REM NULELEMENTER I F OVERSPRINGS
1030 IF F(I)=0 THEN 1230
1040 J=F(I)
1050 F(I)=0
1060 REM HVIS J<0 SVARER J TIL ET TRÆK I UNDERP
1070 IF J<0 THEN 1200
1080 REM HOVED.PROG.TRÆK. HVIS V=1 VAR DET GODT
1090 REM HVIS V=-1 VAR DET ET DARLIGT TRÆK
1100 E(I,J)=E(I,J)+V
1110 GOTO 1230
1200 REM UNDER.PROG.TRÆK. HVIS V=-1 VAR DET GODT
1210 REM HVIS V=1 VAR DET ET DARLIGT TRÆK
1220 E(I,-J)=E(I,-J)-V
1230 NEXT
1300 NEXT
1310 PRINT "AF";N;"SPIL TABES";T
1320 IF T>0 THEN 400
1330 PRINT"JEG HAR VIST LÆRT AT SPILLE NIM"
1340 GOTO 400
2000 REM MODSPILLER
2010 REM P0 ER ET GODT PINDEANTAL I BUNKEN,DER
2020 REM INDEHLD. P PINDE. DER TREKKES K PINDE
2030 P0=1+4*INT((P-1)/4)
2040 K=P-P0
2050 IF K<>0 THEN 2070
2060 K=INT(3*AND(1))+1
2070 RETURN

```

Nip og Nap

Af KNUD HERMANSEN og EJVIND JENSEN

Tegninger: KIRSTEN JENSENIUS



3 VI LÆSER
historier

NYT PÆDAGOGISK FORLAG

KØLLEGÅRDSKOLEN

SÆLGES:

Billig hardcopy fra NASCOM.

En brugt Siemens fjernskriver (gammel model fra P&T).


Pris kr 500,00 afhentet (den er meget tung)

Henvendelse til Knud Christensen

Gormsgade 8 1.tv

2200 København N

Tlf: 01-839442



SÆLGES

Nascom 2 med 32 K ram-A kort
 Micro-case indbygningskasse, strømforsyning
 Kassettebåndoptager + Eprombrønder
 Samt software ad libitum.
 Ring og giv et bud.

M. C. Langholz 02 88 55 62

Print til Morten Kølbaeks Eprombrønder
SÆLGES
 Glasfiber 25,00 kr.
 Komponenter kan eventuelt leveres ekstra.

Henvendelse efter 16.00 til Hans Ole Groth
 09 52 24 56

SÆLGES.

Program til løsning af Rubriks terning.
 Start fra et vilkårligt punkt med
 selvvalgte farver eller tegn.

Kan køre automatisk, eller du kan
 forsøge dig frem efter dine egne ide-
 er.

Terningen vises som den er i NN.7.
 og fremgangsmåden er som i NN.6.
 Basic. Pris pr. bånd (N.2)
 50,00.Kr. Asbjørn Lind.Sidevolden 23
 giro 8 13 50 29.

GEM DIT EKSEMPLAR

af programbiblioteket
 og medlemisten,

DA

vi på dette sted vil opdatere begge lister løbende. Det er således kun
 vores hensigt, at udgive nye liste 1 gang om året.

sir Asbjørn.

ALMINDELIGE OPLYSNINGER OM FORENINGEN :

Bestyrelsens sammensætning:

Formand Asbjørn Lind
 Sidevolden 23
 2730 Herlev
 02 91 71 82

Næstformand Jesper Skavin
 Broholms Alle 3
 2920 Charlottenlund
 01 64 03 14

Kasserer Erik Hansen
 Lyngby Kirkestræde 6.1
 2800 Lyngby
 02 88 60 55 (mellem 8 og 15.30)

Sekretær Carsten Senholt
 Blommevangen 6
 2760 Måløv
 02 66 19 65

 Ole Hasselbalch
 Vibeskrænten 9
 2750 Ballerup
 02 97 70 13

 Søren Sørensen
 Højlundvej 13
 3500 Værløse
 02 48 31 01

 Frank Damgård
 Kastebjergvej 26A
 2750 Ballerup
 02 97 10 20

Henvendelse til foreningen:

Indmeldelse, adresseændringer o.l. til kassereren
Programbibliotek til næstformanden

Øvrige henvendelser til formanden
(herunder annoncer/stof til NASCOM NYT)

Indmeldelsesgebyr: 25,00 kr.
Kontingent 1.7.81 - 1.7.82: 80,00 kr.

Oplag: 250

Redaktionen sluttet den 5.9.81
Husk at gamle numre kan købes hos Ole for 10 kr./stk +porto
Printerservice hos formanden

Annoncepris 0.75 kr. pr. A4 side (siderne 4 - n-2) pr. medlem
Indlevering foreningens adresse.